A2a. Definition of the V-Whittaker function, and various relations

We define $V_{\kappa,s}$ as a linear combination of M-Whittaker functions. The substitution rule **Whsub** expresses both $V_{\kappa,s}$ and $W_{\kappa,s}$ as a linear combination of $M_{\kappa,s}$ and $M_{\kappa,-s}$ (meromorphic in s).

In[ ]:=
```
Clear[WhittakerV, vvw, vvv]
vvw[kp_, s_] = (-Pi) (Sin[2 Pi s])^(-1) Gamma[1/2 - s - kp]^(-1) Gamma[1 + 2 s]^(-1);
vvv[kp_, s_] = Pi I Sin[2 Pi s]^(-1) E^(Pi I s) Gamma[1/2 - s + kp]^(-1) Gamma[1 + 2 s]^(-1);

Clear[Whsub]
Whsub := {WhittakerW[kp_, s_, tau_] :→ vvw[kp, s] WhittakerM[kp, s, tau] +
     vvw[kp, -s] WhittakerM[kp, -s, tau], WhittakerV[kp_, s_, tau_] :→
    vvv[kp, s] WhittakerM[kp, s, tau] + vvv[kp, -s] WhittakerM[kp, -s, tau]}
```

Check of the expression of $M_{\kappa,s}$ in the W- and V-Whittaker functions

In[ ]:=
```
relMWV = WhittakerM[kap, s, tau] ==
  E^(Pi I kap) Gamma[2 s + 1] ( -I E^(-Pi I s) Gamma[1/2 + s + kap]^(-1) WhittakerW[kap, s, tau] -
     Gamma[1/2 + s - kap]^(-1) WhittakerV[kap, s, tau])
```

Out[ ]= WhittakerM[kap, s, tau] ==

$$e^{i\,\mathrm{kap}\,\pi}\,\mathrm{Gamma}[1+2\,s]\left(-\frac{\mathrm{WhittakerV}[\mathrm{kap},s,\mathrm{tau}]}{\mathrm{Gamma}\left[\frac{1}{2}-\mathrm{kap}+s\right]}-\frac{i\,e^{-i\,\pi\,s}\,\mathrm{WhittakerW}[\mathrm{kap},s,\mathrm{tau}]}{\mathrm{Gamma}\left[\frac{1}{2}+\mathrm{kap}+s\right]}\right)$$

Mathematica needs more help

In[ ]:=
```
Clear[gmsub]
gmsub[xx_] := {Gamma[1/2 + xx] → Pi Gamma[1/2 - xx]^(-1) Cos[Pi xx]^(-1)};
relMWV //. Whsub // Simplify
% /. gmsub[kap - s] /. gmsub[kap + s] // Simplify
```

Out[ ]=

$$\left(1+i\,e^{i\,\pi\,(\mathrm{kap}-s)}\,\pi\,\mathrm{Csc}[2\,\pi\,s]\left(\frac{e^{2\,i\,\pi\,s}}{\mathrm{Gamma}\left[\frac{1}{2}+\mathrm{kap}-s\right]\mathrm{Gamma}\left[\frac{1}{2}-\mathrm{kap}+s\right]}-\frac{1}{\mathrm{Gamma}\left[\frac{1}{2}-\mathrm{kap}-s\right]\mathrm{Gamma}\left[\frac{1}{2}+\mathrm{kap}+s\right]}\right)\right)\mathrm{WhittakerM}[\mathrm{kap},s,\mathrm{tau}]==0$$

Out[ ]= True